Example: Exercise 23, Section 16.7 (The Flux Integral)

Below, we look at the built-in "Flux" command, and how to do the computation by hand.
We also look at the nice graphics features!

```
> #Define the vector field:
> with(VectorCalculus):  with(Student[VectorCalculus]):
> F:=VectorField(<x*y, y*z, z*x>);
```

$$F := (xy)\bar{e}_x + (yz)\bar{e}_y + (zx)\bar{e}_z \qquad (1)$$

```
> #Define the surface:
> S:=<u, v, 4-u^2-v^2>;
```

$$S := (u)e_x + (v)e_y + (-u^2 - v^2 + 4)e_z \qquad (2)$$

```
> # Plot the flux:
> Flux(F, Surface(S, u=0..1, v=0..1), output=plot);
```



The vector field arrows, the surface through which the field passes, and vectors normal to the surface.

```
> Flux(F, Surface(S, u=0..1, v=0..1));
```

$$\frac{713}{180} \tag{3}$$

```
> Flux(F, Surface(S, u=0..1, v=0..1), output=integral);
```

$$\int_0^1 \int_0^1 (2\,u^2\,v + 2\,v^2\,(-u^2 - v^2 + 4) + (-u^2 - v^2 + 4)\,u)\,dv\,du \tag{4}$$

```
> # Computing the flux "by hand":  First substitute the surface
  functions into the vector field:
> F1:=subs({x=S[1], y=S[2], z=S[3]},F);
```

$$F1 := (u\,v)\bar{e}_x + (v\,(-u^2 - v^2 + 4))\bar{e}_y + ((-u^2 - v^2 + 4)\,u)\bar{e}_z \tag{5}$$

```
> # Find the surface normal vector (not a unit vector):
  N:=CrossProduct(diff(S,u), diff(S,v));
```

$$N := 2\,u e_x + 2\,v e_y + e_z \tag{6}$$

```
> # Compute the flux integral:
  Integrand:=DotProduct(F1,N);
```

$$Integrand := 2\,u^2\,v + 2\,v^2\,(-u^2 - v^2 + 4) + (-u^2 - v^2 + 4)\,u \tag{7}$$

```
> FF:=int(int(Integrand,v=0..1),u=0..1);
```

$$FF := \frac{713}{180} \tag{8}$$

Other tips:

(1) Remember to use the multiplication sign, *, where needed!
(2) The exponential function is not e^x, it is exp(x)